pragma solidity 0.8.4;

// SPDX-License-Identifier: MIT

// File @openzeppelin/contracts/utils/Context.sol@v4.3.2

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File @openzeppelin/contracts/access/Ownable.sol@v4.3.2

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), 'Ownable: caller is not the owner');
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), 'Ownable: new owner is the zero address');
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File @openzeppelin/contracts/token/ERC20/IERC20.sol@v4.3.2

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File @openzeppelin/contracts/utils/Address.sol@v4.3.2

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, 'Address: insufficient balance');

        (bool success, ) = recipient.call{value: amount}('');
        require(success, 'Address: unable to send value, recipient may have reverted');
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, 'Address: low-level call failed');
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, 'Address: low-level call with value failed');
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, 'Address: insufficient balance for call');
        require(isContract(target), 'Address: call to non-contract');

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, 'Address: low-level static call failed');
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), 'Address: static call to non-contract');

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, 'Address: low-level delegate call failed');
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), 'Address: delegate call to non-contract');

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol@v4.3.2

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            'SafeERC20: approve from non-zero to non-zero allowance'
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, 'SafeERC20: decreased allowance below zero');
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, 'SafeERC20: low-level call failed');
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), 'SafeERC20: ERC20 operation did not succeed');
        }
    }
}

// File @uniswap/v2-core/contracts/interfaces/IUniswapV2Pair.sol@v1.0.1

interface IUniswapV2Pair {
    event Approval(address indexed owner, address indexed spender, uint256 value);
    event Transfer(address indexed from, address indexed to, uint256 value);

    function name() external pure returns (string memory);

    function symbol() external pure returns (string memory);

    function decimals() external pure returns (uint8);

    function totalSupply() external view returns (uint256);

    function balanceOf(address owner) external view returns (uint256);

    function allowance(address owner, address spender) external view returns (uint256);

    function approve(address spender, uint256 value) external returns (bool);

    function transfer(address to, uint256 value) external returns (bool);

    function transferFrom(
        address from,
        address to,
        uint256 value
    ) external returns (bool);

    function DOMAIN_SEPARATOR() external view returns (bytes32);

    function PERMIT_TYPEHASH() external pure returns (bytes32);

    function nonces(address owner) external view returns (uint256);

    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    event Mint(address indexed sender, uint256 amount0, uint256 amount1);
    event Burn(address indexed sender, uint256 amount0, uint256 amount1, address indexed to);
    event Swap(
        address indexed sender,
        uint256 amount0In,
        uint256 amount1In,
        uint256 amount0Out,
        uint256 amount1Out,
        address indexed to
    );
    event Sync(uint112 reserve0, uint112 reserve1);

    function MINIMUM_LIQUIDITY() external pure returns (uint256);

    function factory() external view returns (address);

    function token0() external view returns (address);

    function token1() external view returns (address);

    function getReserves()
        external
        view
        returns (
            uint112 reserve0,
            uint112 reserve1,
            uint32 blockTimestampLast
        );

    function price0CumulativeLast() external view returns (uint256);

    function price1CumulativeLast() external view returns (uint256);

    function kLast() external view returns (uint256);

    function mint(address to) external returns (uint256 liquidity);

    function burn(address to) external returns (uint256 amount0, uint256 amount1);

    function swap(
        uint256 amount0Out,
        uint256 amount1Out,
        address to,
        bytes calldata data
    ) external;

    function skim(address to) external;

    function sync() external;

    function initialize(address, address) external;
}

// File @uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router01.sol@v1.1.0-beta.0

interface IUniswapV2Router01 {
    function factory() external pure returns (address);

    function WETH() external pure returns (address);

    function addLiquidity(
        address tokenA,
        address tokenB,
        uint256 amountADesired,
        uint256 amountBDesired,
        uint256 amountAMin,
        uint256 amountBMin,
        address to,
        uint256 deadline
    )
        external
        returns (
            uint256 amountA,
            uint256 amountB,
            uint256 liquidity
        );

    function addLiquidityETH(
        address token,
        uint256 amountTokenDesired,
        uint256 amountTokenMin,
        uint256 amountETHMin,
        address to,
        uint256 deadline
    )
        external
        payable
        returns (
            uint256 amountToken,
            uint256 amountETH,
            uint256 liquidity
        );

    function removeLiquidity(
        address tokenA,
        address tokenB,
        uint256 liquidity,
        uint256 amountAMin,
        uint256 amountBMin,
        address to,
        uint256 deadline
    ) external returns (uint256 amountA, uint256 amountB);

    function removeLiquidityETH(
        address token,
        uint256 liquidity,
        uint256 amountTokenMin,
        uint256 amountETHMin,
        address to,
        uint256 deadline
    ) external returns (uint256 amountToken, uint256 amountETH);

    function removeLiquidityWithPermit(
        address tokenA,
        address tokenB,
        uint256 liquidity,
        uint256 amountAMin,
        uint256 amountBMin,
        address to,
        uint256 deadline,
        bool approveMax,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external returns (uint256 amountA, uint256 amountB);

    function removeLiquidityETHWithPermit(
        address token,
        uint256 liquidity,
        uint256 amountTokenMin,
        uint256 amountETHMin,
        address to,
        uint256 deadline,
        bool approveMax,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external returns (uint256 amountToken, uint256 amountETH);

    function swapExactTokensForTokens(
        uint256 amountIn,
        uint256 amountOutMin,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external returns (uint256[] memory amounts);

    function swapTokensForExactTokens(
        uint256 amountOut,
        uint256 amountInMax,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external returns (uint256[] memory amounts);

    function swapExactETHForTokens(
        uint256 amountOutMin,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external payable returns (uint256[] memory amounts);

    function swapTokensForExactETH(
        uint256 amountOut,
        uint256 amountInMax,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external returns (uint256[] memory amounts);

    function swapExactTokensForETH(
        uint256 amountIn,
        uint256 amountOutMin,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external returns (uint256[] memory amounts);

    function swapETHForExactTokens(
        uint256 amountOut,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external payable returns (uint256[] memory amounts);

    function quote(
        uint256 amountA,
        uint256 reserveA,
        uint256 reserveB
    ) external pure returns (uint256 amountB);

    function getAmountOut(
        uint256 amountIn,
        uint256 reserveIn,
        uint256 reserveOut
    ) external pure returns (uint256 amountOut);

    function getAmountIn(
        uint256 amountOut,
        uint256 reserveIn,
        uint256 reserveOut
    ) external pure returns (uint256 amountIn);

    function getAmountsOut(uint256 amountIn, address[] calldata path) external view returns (uint256[] memory amounts);

    function getAmountsIn(uint256 amountOut, address[] calldata path) external view returns (uint256[] memory amounts);
}

// File @chainlink/contracts/src/v0.8/interfaces/AggregatorV3Interface.sol@v0.2.2

interface AggregatorV3Interface {
    function decimals() external view returns (uint8);

    function description() external view returns (string memory);

    function version() external view returns (uint256);

    // getRoundData and latestRoundData should both raise "No data present"
    // if they do not have data to report, instead of returning unset values
    // which could be misinterpreted as actual reported values.
    function getRoundData(uint80 _roundId)
        external
        view
        returns (
            uint80 roundId,
            int256 answer,
            uint256 startedAt,
            uint256 updatedAt,
            uint80 answeredInRound
        );

    function latestRoundData()
        external
        view
        returns (
            uint80 roundId,
            int256 answer,
            uint256 startedAt,
            uint256 updatedAt,
            uint80 answeredInRound
        );
}

// File contracts/PaymentPortal.sol

contract PaymentPortal is Ownable {
    using SafeERC20 for IERC20;

    uint256 private constant HUNDRED_PERCENT = 1000;
    uint256 private constant UNISWAP_FEE = HUNDRED_PERCENT - 3; // 0.3%

    uint256 private subscriptionDuration = 30 days; // in seconds

    // All payments will be sent to this address
    address private paymentReceiverA;
    address private paymentReceiverB;
    uint256 private paymentShareA = HUNDRED_PERCENT;

    IERC20 private immutable albtToken;
    IERC20 private immutable usdtToken;

    address private immutable uniswapRouter;

    uint256 private priceWithALBT;
    uint256 private priceWithUSDT;

    uint256 private maxSlippage = (HUNDRED_PERCENT * 5) / 100; // 5%

    // Whitelisted addresses
    mapping(address => bool) private whitelist;

    // Stores the timestamp until which each address has access
    mapping(address => uint256) private access;

    // Chainlink price feed aggregator
    AggregatorV3Interface private immutable priceFeed;

    constructor(
        address _paymentReceiverA, // required
        address _paymentReceiverB, // optional, but _paymentShareA must be 1000
        uint256 _paymentShareA, // what percentage of payments will go to payment receiver address A (1000 = 100%)
        address _albtToken, // address of the ALBT token
        address _usdtToken, // address of the USDT token
        address _uniswapRouter, // address of the uniswap router
        uint256 _priceWithALBT, // price in USDT when paying with ALBT (USDT uses 6 decimals)
        uint256 _priceWithUSDT, // price in USDT when paying with USDT (USDT uses 6 decimals)
        address _priceFeed // address of the price feed
    ) {
        _setPaymentReceivers(_paymentReceiverA, _paymentReceiverB);
        _setPaymentShareA(_paymentShareA);

        require(_uniswapRouter != address(0), 'PaymentPortal: Uniswap router address cannot be 0');
        uniswapRouter = _uniswapRouter;

        require(_albtToken != address(0), 'PaymentPortal: ALBT token address cannot be 0');
        require(_usdtToken != address(0), 'PaymentPortal: USDT token address cannot be 0');
        albtToken = IERC20(_albtToken);
        usdtToken = IERC20(_usdtToken);

        require(_priceWithALBT < _priceWithUSDT, 'PaymentPortal: Price with ALBT must be less than price with USDT');
        priceWithALBT = _priceWithALBT;
        priceWithUSDT = _priceWithUSDT;

        require(_priceFeed != address(0), 'PaymentPortal: Price feed address cannot be 0');
        priceFeed = AggregatorV3Interface(_priceFeed);
    }

    function getPaymentReceivers() public view returns (address, address) {
        return (paymentReceiverA, paymentReceiverB);
    }

    function _setPaymentReceivers(address _paymentReceiverA, address _paymentReceiverB) internal {
        require(_paymentReceiverA != address(0), 'PaymentPortal: Payment receiver A must be set');

        require(
            paymentShareA == HUNDRED_PERCENT || _paymentReceiverB != address(0),
            'PaymentPortal: If payment is shared, payment receiver B must be set'
        );

        paymentReceiverA = _paymentReceiverA;
        paymentReceiverB = _paymentReceiverB;
    }

    function setPaymentReceivers(address _paymentReceiverA, address _paymentReceiverB) public onlyOwner {
        _setPaymentReceivers(_paymentReceiverA, _paymentReceiverB);
    }

    function getPaymentShareA() public view returns (uint256) {
        return paymentShareA;
    }

    function _setPaymentShareA(uint256 _paymentShareA) internal {
        require(
            _paymentShareA == HUNDRED_PERCENT || paymentReceiverB != address(0),
            'PaymentPortal: If payment is shared, payment receiver B must be set'
        );

        require(_paymentShareA <= HUNDRED_PERCENT, 'PaymentPortal: Payment share must be between 0% and 100%');

        paymentShareA = _paymentShareA;
    }

    function setPaymentShareA(uint256 _paymentShareA) public onlyOwner {
        _setPaymentShareA(_paymentShareA);
    }

    function setPriceWithALBT(uint256 _priceWithALBT) public onlyOwner {
        require(_priceWithALBT < priceWithUSDT, 'PaymentPortal: Price with ALBT must be less than price with USDT');

        priceWithALBT = _priceWithALBT;
    }

    function setPriceWithUSDT(uint256 _priceWithUSDT) public onlyOwner {
        require(priceWithALBT < _priceWithUSDT, 'PaymentPortal: Price with ALBT must be less than price with USDT');

        priceWithUSDT = _priceWithUSDT;
    }

    function getMaxSlippage() public view returns (uint256) {
        return maxSlippage;
    }

    // Max slippage is a percentage, 10 = 1%, 55 = 5.5%
    function setMaxSlippage(uint256 _maxSlippage) public onlyOwner {
        require(_maxSlippage <= HUNDRED_PERCENT, 'PaymentPortal: Max slippage must be between 0% and 100%');

        maxSlippage = _maxSlippage;
    }

    function getSubscriptionDuration() public view returns (uint256) {
        return subscriptionDuration;
    }

    function setSubscriptionDuration(uint256 _subscriptionDuration) public onlyOwner {
        require(_subscriptionDuration >= 1 weeks, 'PaymentPortal: Subscription duration must be at least 1 week');

        subscriptionDuration = _subscriptionDuration;
    }

    function addToWhitelist(address _wallet) public onlyOwner {
        whitelist[_wallet] = true;
    }

    function removeFromWhitelist(address _wallet) public onlyOwner {
        whitelist[_wallet] = false;
    }

    function isWhitelisted(address _wallet) public view returns (bool) {
        return whitelist[_wallet];
    }

    // Returns the timestamp until which the wallet has access
    function accessTimestamp(address _wallet) public view returns (uint256) {
        return access[_wallet];
    }

    // Checks if the wallet has access, either via the whitelist or by a payment which is still not expired
    function hasAccess(address _wallet) public view returns (bool) {
        return whitelist[_wallet] || access[_wallet] > block.timestamp;
    }

    // Returns the amount of ALBT you need to pay when paying with ALBT, calculated from a discounted USDT rate
    function getPriceInALBT() public view returns (uint256) {
        return calculateUSDTtoALBT(priceWithALBT);
    }

    // Returns the amount of USDT you have to pay when paying with USDT
    function getPriceInUSDT() public view returns (uint256) {
        return priceWithUSDT;
    }

    // When you pay with ALBT the tokens are transfered directly
    function payALBT(address walletToGiveAccess, uint256 maxALBTPayment) public {
        require(!hasAccess(walletToGiveAccess), 'PaymentPortal: Already has access');

        uint256 albtAmount = calculateUSDTtoALBT(priceWithALBT);

        // Needed because USDT or chainlink price can change at any moment
        require(albtAmount <= maxALBTPayment, 'PaymentPortal: Price exceeds max payment');

        transferPayment(msg.sender, albtAmount);

        access[walletToGiveAccess] = block.timestamp + subscriptionDuration;
    }

    // When you pay with USDT the tokens are transfered to this contract, then swapped to ALBT, and then transfered
    function payUSDT(address walletToGiveAccess, uint256 maxUSDTPayment) public {
        require(!hasAccess(walletToGiveAccess), 'PaymentPortal: Already has access');

        // Needed because price can be changed at any moment, this way user can determine how much he's willing to pay
        require(priceWithUSDT <= maxUSDTPayment, 'PaymentPortal: Price exceeds max payment');

        usdtToken.safeTransferFrom(msg.sender, address(this), priceWithUSDT);

        // Expected amount of ALBT to get back, used to calculate slippage
        uint256 albtAmount = calculateUSDTtoALBT(priceWithUSDT);

        address[] memory route = new address[](2);
        route[0] = address(usdtToken);
        route[1] = address(albtToken);

        usdtToken.safeApprove(uniswapRouter, priceWithUSDT);

        uint256[] memory amounts = IUniswapV2Router01(uniswapRouter).swapExactTokensForTokens(
            priceWithUSDT,
            (albtAmount * (HUNDRED_PERCENT - maxSlippage)) / HUNDRED_PERCENT, // max % slippage
            route,
            address(this),
            block.timestamp
        );

        uint256 exactALBTAmount = amounts[amounts.length - 1];

        transferPayment(address(this), exactALBTAmount);

        access[walletToGiveAccess] = block.timestamp + subscriptionDuration;
    }

    // Splits the payment among address A and B based on the share
    function transferPayment(address _from, uint256 _amount) internal {
        uint256 amountA = (_amount * paymentShareA) / HUNDRED_PERCENT;
        uint256 amountB = _amount - amountA;

        if (_from == address(this)) {
            if (amountA > 0) {
                albtToken.safeTransfer(paymentReceiverA, amountA);
            }

            if (amountB > 0) {
                albtToken.safeTransfer(paymentReceiverB, amountB);
            }
        } else {
            if (amountA > 0) {
                albtToken.safeTransferFrom(_from, paymentReceiverA, amountA);
            }

            if (amountB > 0) {
                albtToken.safeTransferFrom(_from, paymentReceiverB, amountB);
            }
        }
    }

    // Calculate the amount of ALBT you will get for an amount of USDT, using the chainlink price feed
    function calculateUSDTtoALBT(uint256 _usdtAmount) internal view returns (uint256) {
        uint8 decimals = priceFeed.decimals();

        (, int256 price, , , ) = priceFeed.latestRoundData();

        // Because we need the price in ALBT decimals, we need to multiply by 10^12 (USDT uses 6 decimals, ALBT uses 18)
        uint256 amountWithoutFees = (_usdtAmount * (10**decimals) * (10**12)) / uint256(price);

        // Uniswap has a 0.3% fee
        return (amountWithoutFees * UNISWAP_FEE) / HUNDRED_PERCENT;
    }
}